NAG Library Routine Document

G13DPF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G13DPF calculates the sample partial autoregression matrices of a multivariate time series. A set of likelihood ratio statistics and their significance levels are also returned. These quantities are useful for determining whether the series follows an autoregressive model and, if so, of what order.

2 Specification

```
 SUBROUTINE G13DPF(K, N, Z, KMAX, M, MAXLAG, PARLAG, SE, QQ, X, PVALUE,
1                  LOGLHD, WORK, LWORK, IWORK, IFAIL)

 INTEGER           K, N, KMAX, M, MAXLAG, LWORK, IWORK(K*M), IFAIL
 double precision  Z(KMAX,N), PARLAG(KMAX,KMAX,M), SE(KMAX,KMAX,M),
1                  QQ(KMAX,KMAX,M), X(M), PVALUE(M), LOGLHD(M),
2                  WORK(LWORK)
```

3 Description

Let $W_t = (w_{1t}, w_{2t}, \ldots, w_{kt})^{\mathrm{T}}$, for $t = 1, 2, \ldots, n$, denote a vector of k time series. The partial autoregression matrix at lag l, P_l, is defined to be the last matrix coefficient when a vector autoregressive model of order l is fitted to the series. P_l has the property that if W_t follows a vector autoregressive model of order p then $P_l = 0$ for $l > p$.

Sample estimates of the partial autoregression matrices may be obtained by fitting autoregressive models of successively higher orders by multivariate least squares; see Tiao and Box (1981) and Wei (1990). These models are fitted using a QR algorithm based on the routines G02DCF and G02DFF. They are calculated up to lag m, which is usually taken to be at most $n/4$.

The routine also returns the asymptotic standard errors of the elements of $\hat{P}_l$ and an estimate of the residual variance-covariance matrix $\hat{\Sigma}_l$, for $l = 1, 2, \ldots, m$. If S_l denotes the residual sum of squares and cross-products matrix after fitting an AR(l) model to the series then under the null hypothesis $H_0 : P_l = 0$ the test statistic

$$X_l = -\big((n - m - 1) - \tfrac{1}{2} - lk\big) \log\left(\frac{|S_l|}{|S_{l-1}|}\right)$$

is asymptotically distributed as χ^2 with k^2 degrees of freedom. X_l provides a useful diagnostic aid in determining the order of an autoregressive model. (Note that $\hat{\Sigma}_l = S_l/(n - l)$.) The routine also returns an estimate of the maximum of the log-likelihood function for each AR model that has been fitted.

4 References

Tiao G C and Box G E P (1981) Modelling multiple time series with applications *J. Am. Stat. Assoc.* **76** 802–816

Wei W W S (1990) *Time Series Analysis: Univariate and Multivariate Methods* Addison–Wesley

5 Parameters

1: K – INTEGER *Input*

On entry: k, the number of time series.

Constraint: K ≥ 1.

2: N – INTEGER *Input*

On entry: n, the number of observations in the time series.

Constraint: N ≥ 4.

3: Z(KMAX,N) – ***double precision*** array *Input*

On entry: Z(i,t) must contain the observation w_{it}, for $i = 1,2,\ldots,k$ and $t = 1,2,\ldots,n$.

4: KMAX – INTEGER *Input*

On entry: the first dimension of the arrays Z, PARLAG, SE and QQ and the second dimension of the arrays PARLAG, SE and QQ as declared in the (sub)program from which G13DPF is called.

Constraint: KMAX $\geq$ K.

5: M – INTEGER *Input*

On entry: m, the number of partial autoregression matrices to be computed. If in doubt set M $= 10$.

Constraint: M ≥ 1 and N $-$ M $-$ (K $\times$ M $+ 1) \geq$ K.

6: MAXLAG – INTEGER *Output*

On exit: the maximum lag up to which partial autoregression matrices (along with their likelihood ratio statistics and their significance levels) have been successfully computed. On a successful exit MAXLAG will equal M. If IFAIL $= 2$ on exit then MAXLAG will be less than M.

7: PARLAG(KMAX,KMAX,M) – ***double precision*** array *Output*

On exit: PARLAG(i,j,l) contains an estimate of the (i,j)th element of the partial autoregression matrix at lag l, $\hat{P}_l(ij)$, for $l = 1,2,\ldots,$MAXLAG, $i = 1,2,\ldots,k$ and $j = 1,2,\ldots,k$.

8: SE(KMAX,KMAX,M) – ***double precision*** array *Output*

On exit: SE(i,j,l) contains an estimate of the standard error of the corresponding element in the array PARLAG.

9: QQ(KMAX,KMAX,M) – ***double precision*** array *Output*

On exit: QQ(i,j,l) contains an estimate of the (i,j)th element of the corresponding variance-covariance matrix $\hat{\Sigma}_l$, for $l = 1,2,\ldots,$MAXLAG, $i = 1,2,\ldots,k$ and $j = 1,2,\ldots,k$.

10: X(M) – ***double precision*** array *Output*

On exit: X(l) contains X_l, the likelihood ratio statistic at lag l, for $l = 1,2,\ldots,$MAXLAG.

11: PVALUE(M) – ***double precision*** array *Output*

On exit: PVALUE(l) contains the significance level of the statistic in the corresponding element of X.

12: LOGLHD(M) – ***double precision*** array *Output*

On exit: LOGLHD(l) contains an estimate of the maximum of the log-likelihood function when an AR(l) model has been fitted to the series, for $l = 1,2,\ldots,$MAXLAG.

13: WORK(LWORK) – ***double precision*** array *Workspace*
14: LWORK – INTEGER *Input*

On entry: the dimension of the array WORK as declared in the (sub)program from which G13DPF is called.

Constraint: LWORK $\geq (k+1)k + l(4+k) + 2l^2$, where $l = mk + 1$.

15: IWORK(K × M) – INTEGER array *Workspace*

16: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, −1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

On exit: IFAIL = 0 unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value −1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value −1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL = 0 or −1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, K < 1,
or N < 4,
or KMAX < K,
or M < 1,
or N − M − (K × M + 1) < K,
or LWORK is too small.

IFAIL = 2

The recursive equations used to compute the sample partial autoregression matrices have broken down at lag MAXLAG + 1. This exit could occur if the regression model is overparameterised. For your settings of k and n the value returned by MAXLAG is the largest permissible value of m for which the model is not overparameterised. All output quantities in the arrays PARLAG, SE, QQ, X, PVALUE and LOGLHD up to and including lag MAXLAG will be correct.

7 Accuracy

The computations are believed to be stable.

8 Further Comments

The time taken is roughly proportional to nmk.

For each order of autoregressive model that has been estimated, G13DPF returns the maximum of the log-likelihood function. An alternative means of choosing the order of a vector AR process is to choose the order for which Akaike's information criterion is smallest. That is, choose the value of l for which $-2 \times \text{LOGLHD}(l) + 2lk^2$ is smallest. You should be warned that this does not always lead to the same choice of l as indicated by the sample partial autoregression matrices and the likelihood ratio statistics.

9 Example

A program to compute the sample partial autoregression matrices of two time series of length 48 up to lag 10.

9.1 Program Text

```
*     G13DPF Example Program Text
*     Mark 16 Release. NAG Copyright 1992.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          KMAX, NMAX, MMAX, LWORK
      PARAMETER        (KMAX=4,NMAX=50,MMAX=10,LWORK=2081)
*     .. Local Scalars ..
      INTEGER          I, IFAIL, J, K, M, MAXLAG, N
*     .. Local Arrays ..
      DOUBLE PRECISION LOGLHD(MMAX), PARLAG(KMAX,KMAX,MMAX),
     +                 PVALUE(MMAX), QQ(KMAX,KMAX,MMAX),
     +                 SE(KMAX,KMAX,MMAX), WORK(LWORK), X(MMAX),
     +                 Z(KMAX,NMAX)
      INTEGER          IWORK(KMAX*MMAX)
*     .. External Subroutines ..
      EXTERNAL         G13DPF, ZPRINT
*     .. Executable Statements ..
      WRITE (NOUT,*) 'G13DPF Example Program Results'
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) K, N, M
*
      IF (K.GT.0 .AND. K.LE.KMAX .AND. N.GE.1 .AND. N.LE.NMAX .AND.
     +    M.GE.1 .AND. M.LE.MMAX) THEN
*
         DO 20 I = 1, K
            READ (NIN,*) (Z(I,J),J=1,N)
   20    CONTINUE
*
         IFAIL = 0
*
         CALL G13DPF(K,N,Z,KMAX,M,MAXLAG,PARLAG,SE,QQ,X,PVALUE,LOGLHD,
     +              WORK,LWORK,IWORK,IFAIL)
*
         CALL ZPRINT(K,N,M,KMAX,MAXLAG,PARLAG,SE,QQ,X,PVALUE,NOUT,IFAIL)
*
      END IF
      END
*
      SUBROUTINE ZPRINT(K,N,M,LDPAR,MAXLAG,PARLAG,SE,QQ,X,PVALUE,NOUT,
     +                  IFAIL)
*
*     .. Scalar Arguments ..
      INTEGER          IFAIL, K, LDPAR, M, MAXLAG, N, NOUT
*     .. Array Arguments ..
      DOUBLE PRECISION PARLAG(LDPAR,LDPAR,M), PVALUE(M),
     +                 QQ(LDPAR,LDPAR,M), SE(LDPAR,LDPAR,M), X(M)
*     .. Local Scalars ..
      DOUBLE PRECISION SUM
      INTEGER          I, I2, J, L
*     .. Local Arrays ..
      CHARACTER*6      ST(6)
*     .. Executable Statements ..
      IF (K.GT.1) WRITE (NOUT,99999)
      IF (K.EQ.1) WRITE (NOUT,99998)
      DO 80 L = 1, MAXLAG
         DO 20 J = 1, K
            SUM = PARLAG(1,J,L)
            ST(J) = '.'
            IF (SUM.GT.1.96D0*SE(1,J,L)) ST(J) = '+'
            IF (SUM.LT.-1.96D0*SE(1,J,L)) ST(J) = '-'
```

```
   20    CONTINUE
         IF (K.EQ.1) THEN
            WRITE (NOUT,99997) L, (PARLAG(1,J,L),J=1,K),
     +        (ST(I2),I2=1,K), QQ(1,1,L), X(L), PVALUE(L)
            WRITE (NOUT,99996) (SE(1,J,L),J=1,K)
         ELSE IF (K.EQ.2) THEN
            WRITE (NOUT,99995) L, (PARLAG(1,J,L),J=1,K),
     +        (ST(I2),I2=1,K), QQ(1,1,L), X(L), PVALUE(L)
            WRITE (NOUT,99994) (SE(1,J,L),J=1,K)
         ELSE IF (K.EQ.3) THEN
            WRITE (NOUT,99993) L, (PARLAG(1,J,L),J=1,K),
     +        (ST(I2),I2=1,K), QQ(1,1,L), X(L), PVALUE(L)
            WRITE (NOUT,99992) (SE(1,J,L),J=1,K)
         ELSE IF (K.EQ.4) THEN
            WRITE (NOUT,99991) L
            WRITE (NOUT,99986) (PARLAG(1,J,L),J=1,K), (ST(I2),I2=1,K),
     +        QQ(1,1,L), X(L), PVALUE(L)
            WRITE (NOUT,99990) (SE(1,J,L),J=1,K)
         END IF
*
         DO 60 I = 2, K
*
            DO 40 J = 1, K
               SUM = PARLAG(I,J,L)
               ST(J) = '.'
               IF (SUM.GT.1.96D0*SE(I,J,L)) ST(J) = '+'
               IF (SUM.LT.-1.96D0*SE(I,J,L)) ST(J) = '-'
   40       CONTINUE
            IF (K.EQ.2) THEN
               WRITE (NOUT,99989) (PARLAG(I,J,L),J=1,K),
     +           (ST(I2),I2=1,K), QQ(I,I,L)
               WRITE (NOUT,99994) (SE(I,J,L),J=1,K)
            ELSE IF (K.EQ.3) THEN
               WRITE (NOUT,99988) (PARLAG(I,J,L),J=1,K),
     +           (ST(I2),I2=1,K), QQ(I,I,L)
               WRITE (NOUT,99992) (SE(I,J,L),J=1,K)
            ELSE IF (K.EQ.4) THEN
               WRITE (NOUT,99987) (PARLAG(I,J,L),J=1,K),
     +           (ST(I2),I2=1,K), QQ(I,I,L)
               WRITE (NOUT,99990) (SE(I,J,L),J=1,K)
            END IF
*
   60    CONTINUE
   80 CONTINUE
*
      WRITE (NOUT,99985) IFAIL
*
      RETURN
*
99999 FORMAT (/' Partial Autoregression Matrices',4X,'Indicator',2X,
     +       'Residual',3X,'Chi-Square',2X,'Pvalue',/37X,'Symbols',3X,
     +       'Variances',3X,'Statistic',/' ---------------------------',
     +       '----',4X,'---------',2X,'---------',2X,'-----------',1X,
     +       '------')
99998 FORMAT (/' Partial Autoregression Function',4X,'Indicator',2X,
     +       'Residual',3X,'Chi-Square',2X,'Pvalue',/37X,'Symbols',3X,
     +       'Variances',3X,'Statistic',/' ---------------------------',
     +       '----',4X,'---------',2X,'---------',2X,'-----------',1X,
     +       '------')
99997 FORMAT (/' Lag',I3,1X,':',F7.3,22X,A1,F14.3,3X,F10.3,F9.3)
99996 FORMAT (10X,'(',F5.3,')')
99995 FORMAT (/' Lag',I3,1X,':',2F8.3,14X,2A1,F13.3,3X,F10.3,F9.3)
99994 FORMAT (11X,'(',F5.3,') (',F5.3,')')
99993 FORMAT (/' Lag',I3,1X,':',3F8.3,6X,3A1,F12.3,3X,F10.3,F9.3)
99992 FORMAT (11X,'(',F5.3,') (',F5.3,') (',F5.3,')')
99991 FORMAT (/' Lag',I3)
99990 FORMAT (3X,'(',F5.3,') (',F5.3,') (',F5.3,') (',F5.3,')')
99989 FORMAT (9X,2F8.3,14X,2A1,F13.3)
```

```
99988 FORMAT (9X,3F8.3,6X,3A1,F12.3)
99987 FORMAT (1X,4F8.3,5X,4A1,F12.3)
99986 FORMAT (1X,4F8.3,5X,4A1,F12.3,3X,F10.3,F9.3)
99985 FORMAT (/' Value of IFAIL parameter on exit from G13DPF = ',I2)
      END
```

9.2 Program Data

```
G13DPF Example Program Data
2 48 10 :  K, no. of series,  N, no. of obs in each series,  M, no. of lags
-1.490 -1.620  5.200  6.230  6.210  5.860  4.090  3.180
 2.620  1.490  1.170  0.850 -0.350  0.240  2.440  2.580
 2.040  0.400  2.260  3.340  5.090  5.000  4.780  4.110
 3.450  1.650  1.290  4.090  6.320  7.500  3.890  1.580
 5.210  5.250  4.930  7.380  5.870  5.810  9.680  9.070
 7.290  7.840  7.550  7.320  7.970  7.760  7.000  8.350
 7.340  6.350  6.960  8.540  6.620  4.970  4.550  4.810
 4.750  4.760 10.880 10.010 11.620 10.360  6.400  6.240
 7.930  4.040  3.730  5.600  5.350  6.810  8.270  7.680
 6.650  6.080 10.250  9.140 17.750 13.300  9.630  6.800
 4.080  5.060  4.940  6.650  7.940 10.760 11.890  5.850
 9.010  7.500 10.020 10.380  8.150  8.370 10.730 12.140 : End of time series
```

9.3 Program Results

```
 G13DPF Example Program Results

 Partial Autoregression Matrices    Indicator  Residual   Chi-Square  Pvalue
                                     Symbols   Variances   Statistic
 -------------------------------    ---------  ---------  ----------- ------

 Lag  1 :   0.757   0.062              +.        2.731       49.884    0.000
          (0.092) (0.092)
           0.061   0.570              .+        5.440
          (0.129) (0.130)

 Lag  2 :  -0.161  -0.135              ..        2.530        3.347    0.502
          (0.145) (0.109)
          -0.093  -0.065              ..        5.486
          (0.213) (0.160)

 Lag  3 :   0.237   0.044              ..        1.755       13.962    0.007
          (0.128) (0.095)
           0.047  -0.248              ..        5.291
          (0.222) (0.165)

 Lag  4 :  -0.098   0.152              ..        1.661        7.071    0.132
          (0.134) (0.099)
           0.402  -0.194              ..        4.786
          (0.228) (0.168)

 Lag  5 :   0.257  -0.026              ..        1.504        5.184    0.269
          (0.141) (0.106)
           0.400  -0.021              ..        4.447
          (0.242) (0.183)

 Lag  6 :  -0.075   0.112              ..        1.480        2.083    0.721
          (0.156) (0.111)
           0.196  -0.106              ..        4.425
          (0.269) (0.192)

 Lag  7 :  -0.054   0.097              ..        1.478        5.074    0.280
          (0.166) (0.121)
           0.574  -0.080              +.        3.838
          (0.267) (0.195)

 Lag  8 :   0.147   0.041              ..        1.415       10.991    0.027
          (0.188) (0.128)
           0.916  -0.242              +.        2.415
          (0.246) (0.167)
```

```
 Lag  9 :  -0.039   0.099             ..        1.322        3.936    0.415
           (0.251) (0.140)
           -0.500   0.173             ..        2.196
           (0.324) (0.181)

 Lag 10 :   0.189   0.131             ..        1.206        3.175    0.529
           (0.275) (0.157)
           -0.183  -0.040             ..        2.201
           (0.371) (0.212)

 Value of IFAIL parameter on exit from G13DPF =  0
```